﻿Shader "Unlit/FogWithNoise"
{
    Properties
    {
        _MainTex ("Texture", 2D) = "white" {}
        _FogDensity("FogDensity（）",Float) = 1.0
        _FogColor("Fog Color",Color) = (1,1,1,1)
        _BottomHeight("Bottom Height",Float) = 0.0
        _TopHeight("Top Height",Float) = 1.0
        _NoiseTex("Noise Texture",2D) = "white"{}
        _FogXSpeed("Fog Horizontal Speed",Float) = 0.1
        _FogYSpeed("Fog Vertical Speed",Float) = 0.1
        _NoiseAmount("Noise Amount",Float) = 1
    }
    SubShader
    {
        CGINCLUDE
            #include "UnityCG.cginc"

            sampler2D _MainTex;
            half4 _MainTex_TexelSize;
            sampler2D _CameraDepthTexture; //Unity传递的 深度纹理
            half _FogDensity;
            fixed4 _FogColor;
            float _BottomHeight;
            float _TopHeight;
            float4x4 _FrustumCorners;
            sampler2D _NoiseTex;
            half _FogXSpeed;
            half _FogYSpeed;
            half _NoiseAmount;

            struct a2v
            {
                float4 vertex : POSITION;
                float2 texcoord : TEXCOORD0;
            };
            struct v2f
            {
                float4 pos : SV_POSITION;
                float2 uv : TEXCOORD0;
                half2 uv_depth : TEXCOORD1; //采样深度纹理的uv
                float4 interpolatedRay : TEXCOORD2; //存储差值后的像素向量
            };
            v2f vert (a2v v)
            {
                v2f o;
                o.pos = UnityObjectToClipPos(v.vertex);
                o.uv = v.texcoord;
                o.uv_depth = v.texcoord;

                #if UNITY_UV_STARTS_AT_TOP //平台差异处理
                if(_MainTex_TexelSize.y<0)
                    o.uv_depth.y = 1 - o.uv_depth.y;
                #endif

                int index = 0;
                if(v.texcoord.x < 0.5 && v.texcoord.y < 0.5)
                {
                    index = 0;
                }
                else if(v.texcoord.x > 0.5 && v.texcoord.y < 0.5)
                {
                    index = 1;
                }
                else if(v.texcoord.x > 0.5 && v.texcoord.y > 0.5)
                {
                    index = 2;
                }
                else
                {
                    index = 3;
                }

                #if UNITY_UV_STARTS_AT_TOP //平台差异处理
                if(_MainTex_TexelSize.y<0)
                    index = 3 - index;
                #endif

                o.interpolatedRay = _FrustumCorners[index];

                return o;
            }
            //基于深度纹理的全局雾效
            fixed4 frag(v2f i) : SV_Target
            {
                //获取深度值（由NDC映射而来）
                float d = SAMPLE_DEPTH_TEXTURE(_CameraDepthTexture,i.uv_depth);
                //获得视角空间下的线性深度值
                float linearDepth = LinearEyeDepth(d);
                //获取世界坐标
                float3 worldPos = _WorldSpaceCameraPos + linearDepth * i.interpolatedRay.xyz;
                //采样Noise
                float2 fogSpeed = _Time.x * float2(_FogXSpeed,_FogYSpeed);
                float noise = (tex2D(_NoiseTex,i.uv + fogSpeed).r - 0.5) * _NoiseAmount;
                //雾效系数（基于高度）
                float fogDensity = (_TopHeight - worldPos.y) / (_TopHeight - _BottomHeight);
                fogDensity = saturate(fogDensity * _FogDensity * (1 + noise));

                fixed4 col = tex2D(_MainTex,i.uv);
                col.rgb = lerp(col.rgb,_FogColor.rgb,fogDensity);

                return col;
            } 
        ENDCG

        ZTest Always Cull Off ZWrite Off
        Pass
        {
            CGPROGRAM
            #pragma vertex vert
            #pragma fragment frag
            ENDCG
        }
    } 
    Fallback Off
}